Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 11
Variation of Parameters

Initializations

```
> restart;
```

11.1 Variation of Parameters

When the right hand side $f(t)$ of the differential equation

$$a\,y'' + b\,y' + c\,y = f(t)$$

does not have a simple differential family, then the method of undetermined coefficients will not apply. Instead we may proceed as follows. Suppose $y_1(t)$ and $y_2(t)$ denote two linearly independent solutions of the corresponding homogeneous equation. Then we try a particular solution of the form

$$y_p(t) = v_1(t)y_1(t) + v_2(t)y_2(t)$$

Compute $y_p'(t)$ as

$$y_p'(t) = v_1'(t)y_1(t) + v_2'(t)y_2(t) + v_1(t)y_1'(t) + v_2(t)y_2'(t)$$

Now let

$$v_1'(t)y_1(t) + v_2'(t)y_2(t) = 0$$

This allows us to simplify $y_p'(t)$ to

$$y_p'(t) = v_1(t)y_1'(t) + v_2(t)y_2'(t)$$

and

$$y_p''(t) = v_1'(t)y_1'(t) + v_2'(t)y_2'(t) + v_1(t)y_1''(t) + v_2(t)y_2''(t)$$

Substitution of $y_p(t)$, $y_p'(t)$, and $y_p''(t)$ into the inhomogeneous differential equation, leads to a second equation for $v_1'(t)$ and $v_2'(t)$. The two equations for $v_1'(t)$ and $v_2'(t)$ can be solved simultaneously and after integration of the result we obtain a desired particular solution of the given inhomogeneous equation.

This method is illustrated in the example below. Mathematical details will be provided in class.

Examples

Example 11.1.1

Solve the initial value problem

$$y'' - 2\,y' + y = t^{-1}e^{t} \qquad y(1) = 1,\ y'(1) = -2$$

Solution
First we determine the roots of the auxiliary equation of the corresponding homogeneous equation.

```
> aux:=r^2-2*r+1=0;
  ev:=solve(aux, r);
```

$$aux := r^2 - 2r + 1 = 0$$

$$ev := 1, 1 \quad \textbf{(2.1.1.1)}$$

We conclude that $y_1(t) = e^t$ and $y_2(t) = te^t$ are two linearly independent solutions of the homogeneous equation. The general solution of the homogeneous equation is given by

$$y_{gh}(t) = c_1 y_1(t) + c_2 y_2(t) = c_1 e^t + c_2 te^t$$

```
> y1:=exp(t);
  y2:=t*exp(t);
  ygh:=c[1]*y1+c[2]*y2;
```

$$y1 := e^t$$

$$y2 := t\,e^t$$

$$ygh := c_1\, e^t + c_2\, t\, e^t \quad \textbf{(2.1.1.2)}$$

To find a particular solution of the inhomogeneous equation we try an expression of the form

$$y_p(t) = v_1(t) y_1(t) + v_2(t) y_2(t)$$

Code this expression and its derivative.

```
> ytry:=v[1](t)*y1+v[2](t)*y2;
  ytryp:=diff(ytry, t);
```

$$ytry := v_1(t)\, e^t + v_2(t)\, t\, e^t$$

$$ytryp := \left(\frac{d}{dt} v_1(t)\right) e^t + v_1(t)\, e^t + \left(\frac{d}{dt} v_2(t)\right) t\, e^t + v_2(t)\, e^t + v_2(t)\, t\, e^t \quad \textbf{(2.1.1.3)}$$

The key to success is to let

$$\left(\frac{d}{dt} v_1(t)\right) e^t + \left(\frac{d}{dt} v_2(t)\right) e^t\, t = 0$$

and use this condition to simplify $y_p{}'(t)$.

```
> eq1:=op(1, ytryp)+op(3, ytryp)=0;
```

$$eq1 := \left(\frac{d}{dt} v_1(t)\right) e^t + \left(\frac{d}{dt} v_2(t)\right) t\, e^t = 0 \quad \textbf{(2.1.1.4)}$$

```
> ytryp:=ytryp-lhs(eq1);
```

$$ytryp := v_1(t)\, e^t + v_2(t)\, e^t + v_2(t)\, t\, e^t \quad \textbf{(2.1.1.5)}$$

Differentiate this result to compute $y_p{}''(t)$.

```
> ytrypp:=diff(ytryp, t);
```

$$ytrypp := \left(\frac{d}{dt} v_1(t)\right) e^t + v_1(t)\, e^t + \left(\frac{d}{dt} v_2(t)\right) e^t + 2\, v_2(t)\, e^t + \left(\frac{d}{dt} v_2(t)\right) t\, e^t + v_2(t)\, t\, e^t \quad \textbf{(2.1.1.6)}$$

A second equation for $v_1{}'(t)$ and $v_2{}'(t)$ is obtained by substituting $y_p(t)$, $y_p{}'(t)$, and $y_p{}''(t)$ into the differential equation.

```
> deq:=diff(y(t), t$2)-2*diff(y(t), t)+y(t)=exp(t)/t;
  eq2:=simplify(eval(deq, {y(t)=ytry, diff(y(t), t)=ytryp,
```

```
diff(y(t), t$2)=ytrypp}));
```

$$deq := \frac{d^2}{dt^2} y(t) - 2\left(\frac{d}{dt} y(t)\right) + y(t) = \frac{e^t}{t}$$

$$eq2 := e^t \left(\frac{d}{dt} v_1(t) + \frac{d}{dt} v_2(t) + \left(\frac{d}{dt} v_2(t)\right) t\right) = \frac{e^t}{t} \quad \textbf{(2.1.1.7)}$$

Now solve the equations **eq1** and **eq2** for $v_1'(t)$ and $v_2'(t)$.

```
> soldv:=solve({eq1, eq2}, {diff(v[1](t), t), diff(v[2](t), t)});
```

$$soldv := \left\{\frac{d}{dt} v_1(t) = -1, \frac{d}{dt} v_2(t) = \frac{1}{t}\right\} \quad \textbf{(2.1.1.8)}$$

Integrate these equations to find $v_1(t)$ and $v_2(t)$. Since we need only one particular solution we can take the integration constants equal to zero.

```
> v1:=map(int, soldv[1], t);
  v2:=map(int, soldv[2], t);
```

$$v1 := v_1(t) = -t$$

$$v2 := v_2(t) = \ln(t) \quad \textbf{(2.1.1.9)}$$

Hence, a particular solution is given by

```
> yp:=eval(ytry, {v1, v2});
```

$$yp := -t\,e^t + \ln(t)\,t\,e^t \quad \textbf{(2.1.1.10)}$$

and a general solution of the inhomogeneous equation is of the form

```
> yg:=ygh+yp;
```

$$yg := c_1 e^t + c_2 t\,e^t - t\,e^t + \ln(t)\,t\,e^t \quad \textbf{(2.1.1.11)}$$

Finally, we implement the initial conditions and determine the constants c_1 and c_2.

```
> eqc1:=eval(yg, t=1)=1;
  eqc2:=eval(diff(yg, t), t=1)=-2;
  val_c:=solve({eqc1, eqc2}, {c[1], c[2]});
```

$$eqc1 := c_1 e + c_2 e - e = 1$$

$$eqc2 := c_1 e + 2 c_2 e - e = -2$$

$$val\_c := \left\{c_1 = \frac{4 + e}{e}, c_2 = -\frac{3}{e}\right\} \quad \textbf{(2.1.1.12)}$$

The solution to the given initial value problem is

```
> ans:=y(t)=eval(yg, val_c);
```

$$ans := y(t) = \frac{(4 + e)\,e^t}{e} - \frac{3\,t\,e^t}{e} - t\,e^t + \ln(t)\,t\,e^t \quad \textbf{(2.1.1.13)}$$